Higher-Order Interactions in Phase Oscillator Networks through Phase Reductions of Oscillators with Phase Dependent Amplitude

Christian Bick, Tobias Böhle & Christian Kuehn

This Mathematica Code is part of the supplementary material for the Paper "Higher-Order Interactions in Phase Oscillator Networks through Phase Reductions of Oscillators with Phase Dependent Amplitude", written by Christian Bick, Tobias Böhle & Christian Kuehn.

Initializations

First, add relevant scripts that we will use, define parameters, variables and the differential equation that describes $R^{(1,*)}(\varphi)$

```
In[]:= Num = 3; (*This is the number of oscillators. The more oscillators,
the longer the calculation takes*)
k = 1;
uniquelistphi = Table[Unique["x"], {Num}];
For[i = 1, i ≤ Num, i++, ϕ[i] = uniquelistphi[[i]];];
phis = Array[ϕ, Num];
g[p_] := Sin[p];
(* Gk[phis/. List→ Sequence] = Sum[(1+δ g[ϕ[l]])/(1+δ g[ϕ[k]]) Cos[ϕ[l]-ϕ[k]+α]-Cos[α] -
   δ g'[ϕ[k]]/(1+δ g[ϕ[k]])((1+δ g[ϕ[l]])/(1+δ g[ϕ[k]])Sin[ϕ[l]-ϕ[k]+α]-Sin[α]), {l,1,Num}]; *)
diffeqalldexp =
  m (1 + δ g[ϕ[k]])^2 * Sum[δ^i * R[i][phis /. List → Sequence], {i, 0, 3}] + 1 / Num *
     Sum[(1 + δ g[ϕ[l]]) / (1 + δ g[ϕ[k]]) Cos[ϕ[l] - ϕ[k] + α] - Cos[α] - δ g '[ϕ[k]] * ((1 + δ g[ϕ[l]]) /
          (1 + δ g[ϕ[k]])^2 Sin[ϕ[l] - ϕ[k] + α] - Sin[α] / (1 + δ g[ϕ[k]])), {l, 1, Num}] ==
   ω * Sum[D[Sum[δ^i * R[i][phis /. List → Sequence], {i, 0, 3}], ϕ[l]], {l, 1, Num}];
```

Next, we define a Rule, that can be used to replace large sums of trigonometric functions with shorter sums, that use order parameters. For example the formula $\frac{1}{\text{Num}}\sum_{j=1}^{\text{Num}}\sin(\varphi_j-\varphi_k)$ can be written as $R[1]\sin(\text{ang}[1]-\varphi_k)$ where $R[1]\,e^{i\,\text{ang}[1]}=\frac{1}{\text{Num}}\sum_{j=1}^{\text{Num}}\sin(\varphi_j)$.

In[∘]:=
```
OrderParamRepRule[arg_, fun_, sumarg_] :=
 Module[{ret},
  ret = arg /. {Plus → List};
  ret = {ret} // Flatten;
  ret = ReplaceAll[ret, {Times → List}];
  ret = Replace[ret, ss_Symbol → {1, ss}, {1}];
  ret = Replace[ret, ss_ang → {1, ss}, {1}];
  pref = 1;
  For[i = 1, i ≤ Length[ret], i++,
   If[ret[[i, 2]] == sumarg,
     int = Abs[ret[[i, 1]]];
     ret[[i, 1]] = ret[[i, 1]] / int;
     ret[[i, 2]] = ang[int];
     pref = pref * r[int];
    ];
  ];
  ret = Map[Function[l, Apply[Times, l]], ret];
  ret = Apply[Plus, ret];
  pref * fun[ret]]
```

Solving the PDEs

The next cell contains the actual calculation process. It calculates the solution of the PDE for $R_k{}^{(1,0)}(\varphi)$, $R_k{}^{(1,1)}(\varphi)$ and $R_k{}^{(1,2)}(\varphi)$. The first part in the for loop solves the PDE, the second part checks if the boundary conditions are satisfied and the third part splits the solutions into sums of the form $R_k{}^{(1,0)}(\varphi)=\frac{1}{\text{fct}[0]}\sum_{l=1}^{\text{Num}}s[0][\varphi_k,\ \varphi_l]$ and calculates s[0], s[1] and s[2]. On each order, the output will be: The solution to the PDE on the respective order in δ, the boundary conditions (if they are satisfied, the output should be an array of zeros), and the summands s[0], s[1] and s[2], respectively.

```
In[∘]:= fct[0] = Num * m; fct[1] = 2 * Num * (m^2 + ω^2);
fct[2] = -4 * Num * m (m^4 + 5 * m^2 ω^2 + 4 ω^4);
(*These are prefactors, that will simplify the output*)
Off[Function::fpct];
For[dord = 0, dord ≤ 2, dord++,
  Print["Order " <> ToString[dord] <> " in δ
      ==========================================================================
      =========================="];
  (*Solve PDE*);
  Prevsols[dord] = Table[sol[i][[1, 1]], {i, 0, dord - 1}];
  eq[dord] = D[diffeqalldexp /. Prevsols[dord], {δ, dord}] /. δ → 0;
  sol[dord] = DSolve[eq[dord], R[dord], phis] /. C1[x__] → 0 ;
  solVal[dord] = R[dord] /. sol[dord][[1]];
  RSol[dord][p__] = emptybody /. emptybody → solVal[dord][p];
  Print[sol[dord]];
  (*Check boundary conditions*);
  bcs[dord] = Table[RSol[dord][phis /. List → Sequence] - RSol[dord][
      phis + 2 π UnitVector[Num, i] /. List → Sequence] // Simplify, {i, 1, Num}];
  Print[bcs[dord]];
  (*Split into Sum*);
  f[dord] =
   (fct[dord] * Integrate[D[RSol[dord][phis /. List → Sequence], ϕ[2]], ϕ[2]]) // Simplify //
    Expand ;
  temp[dord][xk_, xl_] = f[dord] /. {ϕ[1] → xk, ϕ[2] → xl};
  a[dord] = 1 / Num * (fct[dord] * RSol[dord][phis /. List → Sequence] -
      Sum[temp[dord][ϕ[k], ϕ[l]], {l, 1, Num}]) // Simplify;
  s[dord][xk_, xl_] = TrigReduce[temp[dord][xk, xl] + a[dord] /. {ϕ[1] → xk, ϕ[2] → xl}];
  Print[s[dord][xk, xl]];
 ];
```

Order 0 in δ
    ==
    =====

$$\left\{\left\{R[0] \to \text{Function}\left[\{x11, x12, x13\}, -\frac{\text{Cos}[x11 - x12 - \alpha] + \text{Cos}[x11 - x13 - \alpha] - 2\,\text{Cos}[\alpha] - 3\, e^{\frac{m\, x11}{\omega}}\; m\, 0}{3\, m}\right]\right\}\right\}$$

{0, 0, 0}

$$-\mathrm{Cos}[xk - xl - \alpha] + \mathrm{Cos}[\alpha]$$

Order 1 in δ

==
=====

$$\Big\{\Big\{\mathrm{R}[1] \to \mathrm{Function}\Big[\{x11,\ x12,\ x13\},\ \frac{1}{6\,(m^2+\omega^2)}\Big(-4\,\omega\,\mathrm{Cos}[x11-\alpha] - \omega\,\mathrm{Cos}\big[x11-2\,x12-\alpha\big] + 2\,\omega\,\mathrm{Cos}\big[2\,x11-x12-\alpha\big] - \omega\,\mathrm{Cos}\big[x11-2\,x13-\alpha\big] + 2\,\omega\,\mathrm{Cos}\big[2\,x11-x13-\alpha\big] - 6\,\omega\,\mathrm{Cos}[x11+\alpha] + 4\,\omega\,\mathrm{Cos}[x12+\alpha] + 4\,\omega\,\mathrm{Cos}[x13+\alpha] - 4\,m\,\mathrm{Sin}[x11-\alpha] + m\,\mathrm{Sin}\big[x11-2\,x12-\alpha\big] + 2\,m\,\mathrm{Sin}\big[2\,x11-x12-\alpha\big] + m\,\mathrm{Sin}\big[x11-2\,x13-\alpha\big] + 2\,m\,\mathrm{Sin}\big[2\,x11-x13-\alpha\big] - 6\,m\,\mathrm{Sin}[x11+\alpha] + 4\,m\,\mathrm{Sin}[x12+\alpha] + 4\,m\,\mathrm{Sin}[x13+\alpha] + 6\,e^{\frac{m\,x11}{\omega}}\,(m^2+\omega^2)\,0\Big)\Big]\Big\}\Big\}$$

{0, 0, 0}

$$-2\,\omega\,\mathrm{Cos}[xk-\alpha] - \omega\,\mathrm{Cos}\big[xk-2\,xl-\alpha\big] + 2\,\omega\,\mathrm{Cos}\big[2\,xk-xl-\alpha\big] - 3\,\omega\,\mathrm{Cos}[xk+\alpha] + 4\,\omega\,\mathrm{Cos}[xl+\alpha] - 2\,m\,\mathrm{Sin}[xk-\alpha] + m\,\mathrm{Sin}\big[xk-2\,xl-\alpha\big] + 2\,m\,\mathrm{Sin}\big[2\,xk-xl-\alpha\big] - 3\,m\,\mathrm{Sin}[xk+\alpha] + 4\,m\,\mathrm{Sin}[xl+\alpha]$$

Order 2 in δ

==
=====

$$\Big\{\Big\{\mathrm{R}[2] \to \mathrm{Function}\Big[\{x11,\ x12,\ x13\},\ \frac{1}{6\,(m^2+\omega^2)\,(m^2+4\,\omega^2)}\Big(-2\,m\,(m^2-5\,\omega^2)\,\mathrm{Cos}\big[2\,x11-\alpha\big] + m^3\,\mathrm{Cos}\big[2\,x11-2\,x12-\alpha\big] + 4\,m\,\omega^2\,\mathrm{Cos}\big[2\,x11-2\,x12-\alpha\big] - 6\,m^3\,\mathrm{Cos}[x11-x12-\alpha] - 24\,m\,\omega^2\,\mathrm{Cos}[x11-x12-\alpha] + m^3\,\mathrm{Cos}\big[3\,x11-x12-\alpha\big] - 5\,m\,\omega^2\,\mathrm{Cos}\big[3\,x11-x12-\alpha\big] + m^3\,\mathrm{Cos}\big[2\,x11-2\,x13-\alpha\big] + 4\,m\,\omega^2\,\mathrm{Cos}\big[2\,x11-2\,x13-\alpha\big] - 6\,m^3\,\mathrm{Cos}[x11-x13-\alpha] - 24\,m\,\omega^2\,\mathrm{Cos}[x11-x13-\alpha] + m^3\,\mathrm{Cos}\big[3\,x11-x13-\alpha\big] - 5\,m\,\omega^2\,\mathrm{Cos}\big[3\,x11-x13-\alpha\big] + 10\,m^3\,\mathrm{Cos}[\alpha] + 40\,m\,\omega^2\,\mathrm{Cos}[\alpha] - 6\,m^3\,\mathrm{Cos}\big[2\,x11+\alpha\big] + 12\,m\,\omega^2\,\mathrm{Cos}\big[2\,x11+\alpha\big] + 5\,m^3\,\mathrm{Cos}[x11+x12+\alpha] - 7\,m\,\omega^2\,\mathrm{Cos}[x11+x12+\alpha] - 2\,m^3\,\mathrm{Cos}\big[2\,x12+\alpha\big] + m\,\omega^2\,\mathrm{Cos}\big[2\,x12+\alpha\big] + 5\,m^3\,\mathrm{Cos}[x11+x13+\alpha] - 7\,m\,\omega^2\,\mathrm{Cos}[x11+x13+\alpha] - 2\,m^3\,\mathrm{Cos}\big[2\,x13+\alpha\big] + m\,\omega^2\,\mathrm{Cos}\big[2\,x13+\alpha\big] + 8\,m^2\,\omega\,\mathrm{Sin}\big[2\,x11-\alpha\big] - 4\,\omega^3\,\mathrm{Sin}\big[2\,x11-\alpha\big] + m^2\,\omega\,\mathrm{Sin}\big[2\,x11-2\,x12-\alpha\big] + 4\,\omega^3\,\mathrm{Sin}\big[2\,x11-2\,x12-\alpha\big] - 2\,m^2\,\omega\,\mathrm{Sin}[x11-x12-\alpha] - 8\,\omega^3\,\mathrm{Sin}[x11-x12-\alpha] - 4\,m^2\,\omega\,\mathrm{Sin}\big[3\,x11-x12-\alpha\big] + 2\,\omega^3\,\mathrm{Sin}\big[3\,x11-x12-\alpha\big] + m^2\,\omega\,\mathrm{Sin}\big[2\,x11-2\,x13-\alpha\big] + 4\,\omega^3\,\mathrm{Sin}\big[2\,x11-2\,x13-\alpha\big] - 2\,m^2\,\omega\,\mathrm{Sin}[x11-x13-\alpha] - 8\,\omega^3\,\mathrm{Sin}[x11-x13-\alpha] - 4\,m^2\,\omega\,\mathrm{Sin}\big[3\,x11-x13-\alpha\big] + 2\,\omega^3\,\mathrm{Sin}\big[3\,x11-x13-\alpha\big] - 2\,m^2\,\omega\,\mathrm{Sin}[\alpha] - 8\,\omega^3\,\mathrm{Sin}[\alpha] + 18\,m^2\,\omega\,\mathrm{Sin}\big[2\,x11+\alpha\big] - 14\,m^2\,\omega\,\mathrm{Sin}[x11+x12+\alpha] - 2\,\omega^3\,\mathrm{Sin}[x11+x12+\alpha] + 5\,m^2\,\omega\,\mathrm{Sin}\big[2\,x12+\alpha\big] + 2\,\omega^3\,\mathrm{Sin}\big[2\,x12+\alpha\big] - 14\,m^2\,\omega\,\mathrm{Sin}[x11+x13+\alpha] - 2\,\omega^3\,\mathrm{Sin}[x11+x13+\alpha] + 5\,m^2\,\omega\,\mathrm{Sin}\big[2\,x13+\alpha\big] + 2\,\omega^3\,\mathrm{Sin}\big[2\,x13+\alpha\big] + 6\,e^{\frac{m\,x11}{\omega}}\,(m^4+5\,m^2\,\omega^2+4\,\omega^4)\,0\Big)\Big]\Big\}\Big\}$$

{0, 0, 0}

```
2 m^4 Cos[2 xk - α] - 10 m^2 ω^2 Cos[2 xk - α] - 2 m^4 Cos[2 xk - 2 xl - α] - 8 m^2 ω^2 Cos[2 xk - 2 xl - α] +
 12 m^4 Cos[xk - xl - α] + 48 m^2 ω^2 Cos[xk - xl - α] - 2 m^4 Cos[3 xk - xl - α] + 10 m^2 ω^2 Cos[3 xk - xl - α] -
 10 m^4 Cos[α] - 40 m^2 ω^2 Cos[α] + 6 m^4 Cos[2 xk + α] - 12 m^2 ω^2 Cos[2 xk + α] - 10 m^4 Cos[xk + xl + α] +
 14 m^2 ω^2 Cos[xk + xl + α] + 4 m^4 Cos[2 xl + α] - 2 m^2 ω^2 Cos[2 xl + α] - 8 m^3 ω Sin[2 xk - α] + 4 m ω^3 Sin[2 xk - α] -
 2 m^3 ω Sin[2 xk - 2 xl - α] - 8 m ω^3 Sin[2 xk - 2 xl - α] + 4 m^3 ω Sin[xk - xl - α] + 16 m ω^3 Sin[xk - xl - α] +
 8 m^3 ω Sin[3 xk - xl - α] - 4 m ω^3 Sin[3 xk - xl - α] + 2 m^3 ω Sin[α] + 8 m ω^3 Sin[α] - 18 m^3 ω Sin[2 xk + α] +
 28 m^3 ω Sin[xk + xl + α] + 4 m ω^3 Sin[xk + xl + α] - 10 m^3 ω Sin[2 xl + α] - 4 m ω^3 Sin[2 xl + α]
```

Next, we define the summands ssin[1] and scos[1] that appear in the solution formula of $R_k^{(1,1)}(\varphi)$ when $g(\phi) = \sin(n\phi)$ and $g(\varphi) = \cos(n\varphi)$. The validity of these functions can be checked with the code above. When you do that, remember to adjust the fct[1] to 2*Num*(m^2 + (nω)^2). They are only relevant for supplementary calculations in the appendix in the paper.

```
In[∘]:= (*gensols def*)
ssin[1][xk_, xl_] :=
  n ω ((n - 2) Cos[n xk - α] - Cos[xk - (n + 1) xl - α] - (n - 3) Cos[(n + 1) xk - xl - α] -
      Cos[xk + (n - 1) xl - α] - (n + 2) Cos[n xk + α] + (n + 3) Cos[(n - 1) xk + xl + α]) +
   m ((n - 2) Sin[n xk - α] + Sin[xk - (n + 1) xl - α] - (n - 3) Sin[(n + 1) xk - xl - α] -
      Sin[xk + (n - 1) xl - α] - (n + 2) Sin[n xk + α] + (n + 3) Sin[(n - 1) xk + xl + α]);
scos[1][xk_, xl_] :=
  n ω (-(n - 2) Sin[n xk - α] - Sin[xk - (n + 1) xl - α] + (n - 3) Sin[(n + 1) xk - xl - α] +
      Sin[xk + (n - 1) xl - α] + (n + 2) Sin[n xk + α] - (n + 3) Sin[(n - 1) xk + xl + α]) +
   m ((n - 2) Cos[n xk - α] - Cos[xk - (n + 1) xl - α] - (n - 3) Cos[(n + 1) xk - xl - α] -
      Cos[xk + (n - 1) xl - α] - (n + 2) Cos[n xk + α] + (n + 3) Cos[(n - 1) xk + xl + α]);
```

Phase Reductions

Given the solutions of these PDEs, we can now assemble the right-hand sides of phase reduced systems. On first order in K most expressions are short, so they can directly be implemented in Matlab. However, on second order, formulas get too long, so we calculate an expression for the formula here in Mathematica and convert the result to a Matlab code. We use a “Mathematica Expression to Matlab m-file Converter” https://library.wolfram.com/infocenter/MathSource/577/ (November 29, 2022) to convert the Mathematica Code to Matlab. The resulting strings can then be copied directly into Matlab. Use the above link to download the files and put them in the same path as this Mathematica notebook. The following Cell then includes the files.

```
In[∘]:= AppendTo[$Path, NotebookDirectory[]];
<< ToMatlab.m
```

Straight-Forward evaluation for few oscillators

If the number of oscillators is small, a straight-forward approach is best.
The result of the following cell was used in the Matlab file `rhs_K2d1_slow.m` and in `rhs_K2d2_slow.m`

```
In[∘]:= ex = s[1][xk, xl] /. {xk → ph, xl → transpose[ph], α → alpha, ω → omega};
ToMatlab[ex]
```

The result of the following cell was used in the Matlab file `rhs_K2d2_slow.m`

```
In[∘]:= ex = s[2][xk, xl] /. {xk → ph, xl → transpose[ph], α → alpha, ω → omega};
ToMatlab[ex]
```

Fast Evaluation for a large number of oscillators

If the number of oscillators is large, we recommend using the order parameter to increase the performance of the code.

Terms of order δ^0

The following cell calculates an expression for $P_{\breve{A}}^{(2,0)}(\phi) = \nabla_u H_{\breve{A}}^{(-,0)}(1, \phi) \cdot R_{\breve{A}}^{(1,0)}(\phi)$. This was used in the Matlab file `rhs_K2d0.m`

```
In[∘]:= ex = 2 Sin[xl - xk + α] * (s[0][xl, xi] - s[0][xk, xi]) // TrigReduce;
ex1 = ex /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xi],
     Sin[x_] :→ OrderParamRepRule[x, Sin, xi]};
ex2 = ex1 /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xl],
     Sin[x_] :→ OrderParamRepRule[x, Sin, xl]};
ToMatlab[ex2 /. {xk → ph, α → alpha, ω → omega}]
```

First order in δ

Next, we calculate $P_{\breve{A}}^{(2,1)}(\phi)$, which consists of two parts. The following cell calculates an expression for the first part $\nabla_u H_{\breve{A}}^{(-,0)}(1, \phi) \cdot R^{(1,1)}(\phi)$

```
In[∘]:= ex = 2 Sin[xl - xk + α] * (s[1][xl, xi] - s[1][xk, xi]) // TrigReduce;
ex1 = ex /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xi],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xi]};
ex2 = ex1 /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xl],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xl]};
ToMatlab[ex2 /. {xk → ph, α → alpha, ω → omega}]
```

The next cell calculates the second part, which is $\nabla_u H_{\breve{A}}^{(-,1)}(1, \phi) \cdot R^{(1,0)}(\phi)$. These two parts are used in the Matlab file `rhs_K2d1.m`

```
In[∘]:= ex = 4 (Sin[xl] - Sin[xk]) * Sin[xl - xk + α] * (s[0][xl, xi] - s[0][xk, xi]) // TrigReduce;
ex1 = ex /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xi],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xi]};
ex2 = ex1 /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xl],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xl]};
ToMatlab[ex2 /. {xk → ph, α → alpha, ω → omega}]
```

Second order in δ

Next, we calculate an expression $P_{\breve{A}}^{(2,2)}(\phi)$, which consists of three parts. All three parts are used in the Matlab file `rhs_K2d2.m`. The first part is given by $\nabla_u H_{\breve{A}}^{(-,0)}(1, \phi) \cdot R^{(1,2)}(\phi)$, which is calculated in the following cell

```
In[∘]:= ex = 2 Sin[xl - xk + α] * (s[2][xl, xi] - s[2][xk, xi]) // TrigReduce;
ex1 = ex /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xi],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xi]};
ex2 = ex1 /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xl],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xl]};
ToMatlab[ex2 /. {xk → ph, α → alpha, ω → omega}]
```

Next, we calculate an expression for $\nabla_u H_{\breve{A}}^{(-,1)}(1, \phi) \cdot R^{(1,1)}(\phi)$

```
In[∘]:= ex = 4 (Sin[xl] - Sin[xk]) * Sin[xl - xk + α] * (s[1][xl, xi] - s[1][xk, xi]) // TrigReduce ;
ex1 = ex /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xi],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xi]};
ex2 = ex1 /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xl],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xl]};
ToMatlab[ex2 /. {xk → ph, α → alpha, ω → omega}]
```

Finally, we calculate an expression for $\nabla_u H_{\tilde{A}}^{(-,2)}(1, \phi) \cdot R^{(1,0)}(\phi)$

```
In[∘]:= ex = 8 * Sin[xk] * (Sin[xk] - Sin[xl]) * Sin[xl - xk + α] * (s[0][xl, xi] - s[0][xk, xi]) // TrigReduce;
ex1 = ex /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xi],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xi]};
ex2 = ex1 /. {Cos[x_] :→ OrderParamRepRule[x, Cos, xl],
    Sin[x_] :→ OrderParamRepRule[x, Sin, xl]};
ToMatlab[ex2 /. {xk → ph, α → alpha, ω → omega}]
```

Dynamics

To investigate the multipliers of the Poincaré return map of periodic orbits in phase reduced systems, we need to define phase reduced systems. The following cell defines the function $H(1, \phi)$ and $\nabla_u H(1, \phi)$, which is required for phase reductions of order 1 and 2 in K.

```
In[∘]:= H = 1 / Num * Table[Sum[(1 + δ g[ϕ[ll]]) / (1 + δ g[ϕ[kk]]) Sin[ϕ[ll] - ϕ[kk] + α] - Sin[α], {ll, 1, Num}],
    {kk, 1, Num}]; (*radii are fixed to 1, but phases are still free arguments*)
gradH = 1 / Num *
   Table[(1 + δ g[ϕ[ll]]) / (1 + δ g[ϕ[kk]]) * Sin[ϕ[ll] - ϕ[kk] + α], {kk, 1, Num}, {ll, 1, Num}];
gradH = gradH - DiagonalMatrix[Diagonal[gradH]];
gradH = gradH - DiagonalMatrix[Total[gradH, {2}]];
```

Next, we define the different parts of a phase reduction. If you want to study a phase reduction with orders up to two in δ and K, run the following cell.

```
In[∘]:= PK1dall = Series[H , {δ, 0, 2}] // Normal;
PK1d0 = SeriesCoefficient[H, {δ, 0, 0}];
PK1d1 = SeriesCoefficient[H, {δ, 0, 1}];
PK1d2 = SeriesCoefficient[H, {δ, 0, 2}];
PK2d0 = Table[Sum[(gradH[[kk, ll]] /. δ → 0) * s[0][ϕ[ll], ϕ[ii]] / fct[0],
    {ll, 1, Num}, {ii, 1, Num}], {kk, 1, Num}];
PK2d1 = Table[Sum[SeriesCoefficient[gradH[[kk, ll]], {δ, 1, 1}] * s[0][ϕ[ll], ϕ[ii]] / fct[0] +
      (gradH[[kk, ll]] /. δ → 0) * s[1][ϕ[ll], ϕ[ii]] / fct[1],
    {ll, 1, Num}, {ii, 1, Num}], {kk, 1, Num}];
PK2d2 = Table[Sum[SeriesCoefficient[gradH[[kk, ll]], {δ, 0, 0}] * s[2][ϕ[ll], ϕ[ii]] / fct[2] +
      SeriesCoefficient[gradH[[kk, ll]], {δ, 1, 1}] * s[1][ϕ[ll], ϕ[ii]] / fct[1] +
      SeriesCoefficient[gradH[[kk, ll]], {δ, 2, 2}] * s[0][ϕ[ll], ϕ[ii]] / fct[0],
    {ll, 1, Num}, {ii, 1, Num}], {kk, 1, Num}];
rhs = ω + K * PK1d0 + K * δ * PK1d1 +
   K * δ^2 * PK1d2 + K^2 * PK2d0 + K^2 * δ * PK2d1 + K^2 * δ^2 * PK2d2;
(*this rhs can be adjusted to contain different orders of delta and K*)
```

Alternatively, if you want to consider a (1, ∞)-phase reduction, use the following right-hand side:

```
rhs = ω + K * H;
```

Moreover, since we want to compare phase reductions to the full system, we introduce the full system next:

```
In[∘]:= uniquelistRad = Table[Unique["Rad"], {Num}];
For[i = 1, i ≤ Num, i++, R[i] = uniquelistRad[[i]];];
Rads = Array[R, Num];
FullRHS = Table[m * R[kk]^2 * (R[kk] - 1) (1 + δ g[ϕ[kk]]) +
      K * 1 / Num * Sum[R[ll] (1 + δ g[ϕ[ll]]) / (1 + δ g[ϕ[kk]]) Cos[ϕ[ll] - ϕ[kk] + α] - R[kk] Cos[α] -
          δ g '[ϕ[kk]] (R[ll] (c + δ g[ϕ[ll]]) / (c + δ g[ϕ[kk]])^2 Sin[ϕ[ll] - ϕ[kk] + α] -
             R[kk] Sin[α] / (1 + δ g[ϕ[kk]])), {ll, 1, Num}], {kk, 1, Num}]~Join~
    Table[ω + K / Num * Sum[R[ll] (1 + δ g[ϕ[ll]]) / (R[kk] (1 + δ g[ϕ[kk]])) Sin[ϕ[ll] - ϕ[kk] + α] -
          Sin[α], {ll, 1, Num}], {kk, 1, Num}];
```

Synchronized Orbit

To study multipliers of a synchronized orbit in phase reduced systems, we linearize the the right-hand side `rhs` at a synchronized state γ to obtain a Jacobian matrix. Due commutative property of these matrices (for different values of γ), we can then integrate it over one period of the orbit.

```
Jacobi = D[rhs, {phis}] /. Table[ϕ[i] → γ, {i, 1, Num}] // Simplify;
poincare = MatrixExp[Integrate[Jacobi, {γ, 0, 2 π}] / ω];
mult = Eigenvalues[poincare] (*contains one 1 eigenvalue, that has to be ignored*)
```

When $\delta = 0$, we can change into a rotating coordinate frame, in which each synchronized state is an equilibrium. Then we linearize around a synchronized state in the full system. The resulting Jacobian matrix is denoted by `Jacobifull` or `Jacobifull2` (results should agree). Then, we determine the eigenvalues of this matrix. These are used to infer the multipliers, as described in the Paper.

```
In[∘]:= Jacobifull =
  D[FullRHS /. δ → 0, {{Rads /. List → Sequence, phis /. List → Sequence}}] /.
    Table[R[kk] → 1, {kk, 1, Num}] /. Table[ϕ[kk] → γ, {kk, 1, Num}];
(* mat = ConstantArray[1, {Num, Num}]- Num*IdentityMatrix[Num];
Jacobifull2 = {{m * IdentityMatrix[Num] + K/Num Cos[α] mat, -K/Num Sin[α]mat},
   {K/Num Sin[α]mat, K/Num Cos[α]mat}} // ArrayFlatten; *)
Jacobifull // MatrixForm;
ev = Eigenvalues[Jacobifull];
ev
```

Next, we study the effect of higher-harmonics in $g(\phi)$ on the multipliers of the synchronized orbit in a (2,1)-phase reduction. If the result of the following cell is 0, it is confirmed that multipliers of the Poincaré return map do not depend on higher-harmonics in a (2,1)-phase reduction.

```
In[∘]:= PK2d1gen =
  Table[Sum[(gradH[[kk, ll]] /. δ → 0) * scos[1][ϕ[ll], ϕ[ii]] / (2 * Num * (m ^ 2 + (n ω) ^ 2)),
    {ll, 1, Num}, {ii, 1, Num}], {kk, 1, Num}];
Jacobigen = D[PK2d1gen, {phis}] /. Table[ϕ[i] → ξ, {i, 1, Num}] // Simplify;
Jacobigen // MatrixForm
fun = Jacobigen[[1, 2]] * Num;
Jacobigen - fun * 1 / Num * (ConstantArray[1, {Num, Num}] - Num * IdentityMatrix[Num]) //
  Simplify // MatrixForm
fun
Assuming[Element[n, Integers], Integrate[fun , {ξ, 0, 2 π}]]
```

Splay Orbit

First, we consider the splay orbit in phase reduced systems, when $\delta = 0$.

In[∘]:=
```
Jacobisplay =
  D[rhs, {phis}] /. δ → 0 /. Table[ϕ[i] → ξ + 2 π i / Num, {i, 1, Num}] // Simplify;
Eigenvalues[Jacobisplay] // Simplify
```

Next, we consider the splay orbit in the full system, when $\delta = 0$. However, the output turns out to be too complicated.

```
Rstar = 1 / 2 * (1 + Sqrt[1 + 4 * K * Cos[α] / m]);
Jacobifullsplay =
  D[FullRHS /. δ → 0, {{Rads /. List → Sequence, phis /. List → Sequence}}] /.
    Table[R[kk] → Rstar, {kk, 1, Num}] /. Table[ϕ[kk] → ξ + 2 π kk / Num, {kk, 1, Num}];
Jacobifullsplay // MatrixForm
evals = Eigenvalues[Jacobifullsplay]
```

Further Calculations

This section contains further calculations, that do not belong in any of the above sections. Some of them have only be used for testing purposes.

```
numrep = {K → .1, δ → .1, ξ → .1517, α → π / 2 + 0.05};
Jacobi /. numrep // MatrixForm
expansionK1 // MatrixForm
mult /. numrep
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} -0.00324992 & 0.00162496 & 0.00162496 \\ 0.00162496 & -0.00324992 & 0.00162496 \\ 0.00162496 & 0.00162496 & -0.00324992 \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} \frac{1}{3} K \delta \left((\text{Sin}[x11] - \text{Sin}[x12]) \text{Sin}[x11 - x12 - \alpha] + (\text{Sin}[x11] - \text{Sin}[x13]) \text{Sin}[x11 - x13 - \alpha]\right) - \frac{1}{3} K \left(\text{Sin}[x11 - \right. \\ \frac{1}{3} K \left(-\text{Sin}[x12 - x13 - \alpha] - 2 \text{Sin}[\alpha] + \text{Sin}[x11 - x12 + \alpha]\right) + \frac{1}{3} K \delta \left((\text{Sin}[x12] - \text{Sin}[x13]) \text{Sin}[x12 - x13 - \alpha] + \right. \\ \frac{1}{3} K \left(-2 \text{Sin}[\alpha] + \text{Sin}[x11 - x13 + \alpha] + \text{Sin}[x12 - x13 + \alpha]\right) + \frac{1}{3} K \delta \left((\text{Sin}[x11] - \text{Sin}[x13]) \text{Sin}[x11 - x13 + \alpha] + \right. \end{pmatrix}$$

Out[∘]= {1.0319, 1.0319, 1}

In[∘]:=
```
MatrixExp[a * (ConstantArray[1, {Num, Num}] / Num - IdentityMatrix[Num])] // Eigenvalues
```

Out[∘]= $\{e^{-a}, e^{-a}, 1\}$

In[∘]:=
```
ex = 8 * Sin[xk] (Sin[xk] - Sin[xl]) * Sin[xl - xk + α] * (s[0][xl, xi] - s[0][xk, xi]) // TrigReduce
```

Out[∘]=
```
Sin[xi - xk] - Sin[xi + xk] + Sin[xi - 3 xl] - Sin[xi + 2 xk - 3 xl] - 2 Sin[xi - xk - 2 xl] +
  3 Sin[xi + xk - 2 xl] - Sin[xi + 3 xk - 2 xl] - 3 Sin[xi - xl] + Sin[xi - 2 xk - xl] +
  2 Sin[xi + 2 xk - xl] + 2 Sin[xi - 3 xk + 2 α] - 3 Sin[xi - xk + 2 α] + Sin[xi + xk + 2 α] +
  Sin[xi - xl + 2 α] - Sin[xi - 2 xk - xl + 2 α] - 2 Sin[xi + xl + 2 α] - Sin[xi - 4 xk + xl + 2 α] +
  3 Sin[xi - 2 xk + xl + 2 α] - Sin[xi - 3 xk + 2 xl + 2 α] + Sin[xi - xk + 2 xl + 2 α]
```

In[∘]:=
```
TrigFactor[s[1][xk, xl] /. α → 0]
```

Out[∘]=

$$-4\,\mathrm{Sin}\left[\frac{xk}{2}-\frac{xl}{2}\right]^2\left(2\,\omega\,\mathrm{Cos}[xk]-\omega\,\mathrm{Cos}[xl]+2\,m\,\mathrm{Sin}[xk]-m\,\mathrm{Sin}[xl]\right)$$

In[∘]:=
```
eq = m R101[p1, p2, p3] + 1 / 3 * (Cos[p2 - p1 + a] + Cos[p3 - p1 + a] - Cos[a]) ==
    ω1 D[R101[p1, p2, p3], p1] + ω2 D[R101[p1, p2, p3], p2] + ω3 D[R101[p1, p2, p3], p3];
DSolve[eq, R101[p1, p2, p3], {p1, p2, p3}]
```

Out[∘]=

$$\Big\{\Big\{\mathrm{R101}[p1, p2, p3] \to \frac{\mathrm{Cos}[a]}{3\,m} - \frac{1}{3\left(m^2+(\omega1-\omega3)^2\right)\left(m^2+(\omega1-\omega2)^2\right)}\Big(m\left(m^2+(\omega1-\omega3)^2\right)\mathrm{Cos}[a-p1+p2]+m\left(m^2+(\omega1-\omega2)^2\right)\mathrm{Cos}[a-p1+p3]+m^2\,\omega1\,\mathrm{Sin}[a-p1+p2]+\omega1^3\,\mathrm{Sin}[a-p1+p2]-m^2\,\omega2\,\mathrm{Sin}[a-p1+p2]-\omega1^2\,\omega2\,\mathrm{Sin}[a-p1+p2]-2\,\omega1^2\,\omega3\,\mathrm{Sin}[a-p1+p2]+2\,\omega1\,\omega2\,\omega3\,\mathrm{Sin}[a-p1+p2]+\omega1\,\omega3^2\,\mathrm{Sin}[a-p1+p2]-\omega2\,\omega3^2\,\mathrm{Sin}[a-p1+p2]+m^2\,\omega1\,\mathrm{Sin}[a-p1+p3]+\omega1^3\,\mathrm{Sin}[a-p1+p3]-2\,\omega1^2\,\omega2\,\mathrm{Sin}[a-p1+p3]+\omega1\,\omega2^2\,\mathrm{Sin}[a-p1+p3]-m^2\,\omega3\,\mathrm{Sin}[a-p1+p3]-\omega1^2\,\omega3\,\mathrm{Sin}[a-p1+p3]+2\,\omega1\,\omega2\,\omega3\,\mathrm{Sin}[a-p1+p3]-\omega2^2\,\omega3\,\mathrm{Sin}[a-p1+p3]\Big)+e^{\frac{m\,p1}{\omega1}}\,c_1\left[p2-\frac{p1\,\omega2}{\omega1},\,p3-\frac{p1\,\omega3}{\omega1}\right]\Big\}\Big\}$$

Out[∘]=

$$\Big\{\Big\{\mathrm{R101}[p1, p2] \to \frac{\mathrm{Cos}[a]-\mathrm{Cos}[a-p1+p2]+2\,e^{\frac{m\,p1}{\omega}}\,m\,c_1[-p1+p2]}{2\,m}\Big\}\Big\}$$